

```

1 REM CASSE BRIQUE Par Yann Venance---
10 CLS
20 INK 0,9
30 MODE 0
40 LOCATE 10,10.
50 a=40
60 v=7
70 PRINT" CASSE BRIQUES "
80 SOUND 1,284,a,v
90 SOUND 1,213 ,a/2,v
100 SOUND 1,225,a/2,v
110 SOUND 1,190,a/2,v
120 SOUND 1,213,a/2,v
130 SOUND 1,169,a/2,v
140 SOUND 1,190,a/2,v
150 SOUND 1,159,a/2,v
160 SOUND 1,165,a/2,v
170 SOUND 49,568,a,v
180 SOUND 42,225,a,v
190 SOUND 28,190,a,v
200 SOUND 1,0,200
210 FOR s=0 TO 1900 :NEXT s
220 '----jeu----
330 KEY 139, CHR$(13)+"mode 1:ink 0,1:ink 1,24:paper 0:pen 1"+CHR$(13)
340 REM copyright venance 10/6/1985
350 CLS
360 BORDER 1:INK 0,1:INK 1,26:INK 2,24:INK 3,6
370 SPEED KEY 15,2
380 ENV 1,1,18,0,11,0,10:ENT 1,10,2,2
390 ENV 3,1,0,16,5,-3,2
400 ENV 2,5,3,3,1,-21,22,9,-3,2:ENT -2,10,2,2,5,-7,1,2,11,3,2,-4,8
410 MODE 1
420 MOVE 30,16:DRAWR 0,400,1:MOVE 610,16:DRAWR 0,400,1
430 PEN 14 :LOCATE 3,1:PRINT STRING$(36,143):PEN 2:LOCATE 3,2:PRINT STRING$(36,143):PEN 1:FOR r=5 TO 6:LOCATE 3,r:PRINT STRING$(36,1
43):NEXT r
440 bx=9
450 essai=5:score=0
460 PEN 1:GOSUB 770
470 FOR m=0 TO 1000 :NEXT
480 GOTO 510:REM debut du jeu
490 LOCATE bx,24:PRINT" ";STRING$(4,131);" "
500 RETURN
510 xa=1:ya=1:IF INT(RND*2)=1 THEN xa=-xa:REM pour chaque essai le debut est ici
520 PEN 1:GOSUB 490
530 ORIGIN 0,400
540 x=bx+4:y=11:x1=x:y1=y
550 x1=x+xa:y1=y+ya:REM debut de la boucle principale
560 IF x1=3 OR x1=38 THEN xa=-xa
570 GOSUB 710
580 IF y1=24 AND x1>bx+1 AND x1<bx+6 THEN ya=-ya:y1=y1-2:SOUND 130,44,8,7,1,1:a=((x>bx+5)OR(x<bx+2)):IF a=-1 THEN xa=xa*a:x1=x1+xa:y
1=y1+1
590 IF y1=25 THEN LOCATE x,y:PRINT" ":GOTO 680
600 GOSUB 490
610 t=TEST((16*x1)-1,-(16*y1)-1):IF t<0 THEN ya=-ya:xz=x1:yz=y1:y1=y1+ya:GOSUB 740:IF t=2 THEN score=score +10:GOSUB 770
620 IF t=3 THEN score=score +20:GOSUB 770
630 IF t=1 THEN score=score +5:GOSUB 770
640 IF y1=1 THEN ya=1
650 LOCATE x,y:PRINT " ":LOCATE x1,y1:PRINT CHR$(233):x=x1:y=y1
660 IF y=1 OR x=3 OR x=38 THEN SOUND 129,78,8,7,1,1
670 GOTO 550:REM fin de la boucle principale
680 essai=essai-1:SOUND 132,19,46,12,2,2:IF essai=0 THEN GOTO 750
690 GOSUB 770
700 GOTO 510
710 IF (INKEY(8)=0 OR INKEY(74)=0)AND bx>2 THEN bx=bx-2:RETURN
720 IF (INKEY(1)=0 OR INKEY(75)=0)AND bx<32 THEN bx=bx+2:RETURN
730 RETURN
740 LOCATE xz,yz:PRINT " ":RETURN
750 IF score>hiscore THEN hiscore=score
760 score=0:essai=5:GOSUB 770:GOTO 410
770 SOUND 130,0,20,13,3,0,31:LOCATE 3,25:PRINT"HISCORE";hiscore:LOCATE 18,25:PRINT "SCORE";score:LOCATE 32,25:PRINT "ESSAI";essai:RE
TURN

```